

A Semantical Proof of Consistency for Minimal Propositional Logic in Coq

Felipe Sasdelli¹, Maycon Amaro¹, Rodrigo Ribeiro¹

¹Departamento de Computação – Universidade Federal de Ouro Preto (UFOP)
35400-000 – Ouro Preto – MG – Brazil

{felipe.sasdelli,maycon.amaro}@aluno.ufop.edu.br, rodrigo.ribeiro@ufop.edu.br

Abstract. *Consistency is a key property of any logical system. However, proofs of consistency usually rely on heavy proof theory notions like admissibility of cut. A more semantics based approach to consistency proofs is to explore the correspondence between a logics and its relation with the evaluation in a λ -calculus, known as Curry-Howard isomorphism. In this work, we describe a Coq formalization of consistency of minimal propositional logics using this semantics approach and compare it with the traditional approaches used in proof theory.*

1. Introduction

A crucial property of a logical system is consistency, which states that it does not entails a contradiction. Basically, consistency implies that not all formulas are provable. While having a simple motivation, consistency proofs rely on the well-known admissibility of cut property, which has a complex inductive proof. Gentzen, in his seminal work, gives the first consistency proof of logic by introducing an auxiliar formalism, the sequent calculus, in which consistency is trivial. Next, Gentzen showed that natural deduction system is equivalent to his sequent calculus extended with an additional rule: the cut rule. The final (and hardest) piece of Gentzen’s proof is to show that the cut rule is redundant, i.e., it is admissible. As a consequence, we know something stronger: all propositions provable in the natural deduction system are also provable in the sequent calculus without cut. Since we know that the sequent calculus is consistent, we hence also know that the natural deduction calculus is [Negri et al. 2001].

However, proving the admissibility of cut is not easy, even for simple logics. Proofs of admissibility need nested inductions and we need to be really careful to ensure a decreasing measure on each use of the inductive hypothesis. Such proofs have a heavy syntactic flavor since they recursively manipulate proof tree structures to eliminate cuts. A more semantic based approach relies on interpreting logics as its underlying λ -calculus and prove consistency by using its computation machinery. In this work, we report the Coq formalization of these two approaches and advocate the use of the latter since it result on easy to follow proofs. We organize this work as follows: Section 2 present basic definitions about the logic considered and Section 3 describe the semantics of our logic objects and its consistency proof. Section 4 presents a brief comparision between two consistency proofs and concludes.

The complete formalization was verified using Coq version 8.10.2 and its available on-line [Sasdelli et al. 2020]. For space reasons, we rely on reader’s intuition to explain Coq code fragments. Good introductions to Coq are available elsewhere [Chlipala 2013].

2. Basic Definitions

First, we consider formulas of a minimal fragment of propositional logics which is formed only by the constant *falsum* (\perp) and logic implication (\supset). Following common practice, we denote contexts by a list of formulas. The following Coq snippets declare these concepts.

Inductive $\alpha : \text{Set} :=$ While types for formulas (α) and contexts (Γ) have an immediate interpretation, the previous types miss an important part of propositional logic: variables. We represent variables by an inductive judgement which states the membership of a formula in a context.

| *Falsum* : α

| *Implies* : $\alpha \rightarrow \alpha \rightarrow \alpha$.

Definition $\Gamma := \text{list } \alpha$.

We let notation $\alpha \in \Gamma$ denote an inductive predicate that states that a formula α is an element of context Γ . The rules for variable judgement and its Coq implementation are presented below.

Inductive $\text{var} : \Gamma \rightarrow \alpha \rightarrow \text{Type} :=$

| *Here* : $\forall G p, \text{var } (p :: G) p$

| *There* : $\forall G p p', \text{var } G p \rightarrow \text{var } (p' :: G) p$.

$\frac{}{\alpha \in (\alpha :: \Gamma)} \{Here\}$

$\frac{\alpha \in \Gamma}{\alpha \in (\beta :: \Gamma)} \{There\}$

The first constructor of type *var* specifies that a formula α is in the context $\alpha :: \Gamma$ and the constructor *There* specifies that if a formula α is in Γ then we have $\alpha \in (\beta :: \Gamma)$, for any formula β .

Using the previous definitions, we can implement natural deduction rules for our minimal logic, as presented below.

Inductive $\text{nd} : \Gamma \rightarrow \alpha \rightarrow \text{Type} :=$

| *Id* : $\forall G p,$

$\text{var } G p \rightarrow$

$\text{nd } G p$

| *ExFalsum* : $\forall G p,$

$\text{nd } G \text{Falsum} \rightarrow$

$\text{nd } G p$

| *Implies_I* : $\forall G p p',$

$\text{nd } (p' :: G) p \rightarrow$

$\text{nd } G (\text{Implies } p' p)$

| *Implies_E* : $\forall G p p',$

$\text{nd } G (\text{Implies } p' p) \rightarrow$

$\text{nd } G p' \rightarrow$

$\text{nd } G p$.

$\frac{x \in \Gamma}{\Gamma \vdash x} \{Id\}$

$\frac{\Gamma \vdash \perp}{\Gamma \vdash \alpha} \{Ex\}$

$\frac{\Gamma \cup \{\alpha\} \vdash \beta}{\Gamma \vdash \alpha \supset \beta} \{\supset-I\}$

$\frac{\Gamma \vdash \alpha \supset \beta \quad \Gamma \vdash \alpha}{\Gamma \vdash \beta} \{\supset-E\}$

The first rule (*Id*) stabilishes that any formula in the context is provable and rule *ExFalsum* defines the principle *ex-falsum quod libet* which allow us to prove any formula if we have a deduction of *Falsum*. Rule *Implies_I* specifies that from a deduction of a formula p from a context $p' :: G$, $\text{nd } (p' :: G) p$, we can prove the implication *Implies* $p' p$. The last rule, *Implies_E*, represents the well-known *modus-ponens*, which allows us to deduce a formula p from deductions of *Implies* $p' p$ and p' .

Next section uses the relation between logics and λ -calculus and its evaluation to prove the consistency of minimal logic.

3. Semantics and Consistency

We prove the consistency of logics by exploring its correspondence with the simply typed λ -calculus. We do this by implementing in Coq a well-known idea [Augustsson and Carlsson 1999] for implementing denotational semantics for λ -term in type theory based proof assistants.

We define the denotation of a formula by recursion on its structure. The idea is to associate the empty type (*False*) with the formula *Falsum* and a function type with formula *Implies p1 p2*, as presented next.

```
Program Fixpoint sem_form (p :  $\alpha$ ) : Type :=
  match p with
  | Falsum  $\Rightarrow$  False | Implies p1 p2  $\Rightarrow$  sem_form p1  $\rightarrow$  sem_form p2
  end.
```

Using *sem_form* function, we can define context semantics as tuples of formula semantics as follows:

```
Program Fixpoint sem_ctx (G :  $\Gamma$ ) : Type :=
  match G with
  |  $\emptyset \Rightarrow$  unit | (t :: G')  $\Rightarrow$  sem_form t  $\times$  sem_ctx G'
  end.
```

Function *sem_ctx* recurses over the structure of the input context building right-nested tuple ending with the Coq *unit* type, which is a type with a unique element. Since context are mapped into tuples, variables must be mapped into projections on such tuples. This would allow us to retrieve the value associated with a variable in a context.

```
Program Fixpoint sem_var {G p}(v : var G p) : sem_ctx G  $\rightarrow$  sem_form p :=
  match v with
  | Here  $\Rightarrow$  fun env  $\Rightarrow$  fst env | There v'  $\Rightarrow$  fun env  $\Rightarrow$  sem_var v' (snd env)
  end.
```

Function *sem_var* receives a variable (value of type *var G p*) and a semantics of a context (a value of type *sem_ctx G*) and returns the value of the formula represented by such variable. Whenever the variable is built using constructor *Here*, we just return the first component of the input context semantics and when we have the constructor *There* we just call *sem_var* recursively.

Our next step is to define the semantics of natural deduction proofs. The semantics of proofs is implemented by function *sem_nat_ded* which maps proofs (values of type *nat_ded G p*) and context semantics (values of type *sem_ctx G*) to the value of input proof conclusion (type *sem_form p*). The first case specifies that the semantics of an identity rule proof (constructor *Id*) is just retrieving the value of the underlying variable in the context semantics by calling function *sem_var*. Second case deals with *ExFalsum* rule: we recurse over the proof object *Hf* which will produces a Coq object of type *False*, which

is empty and so we can finish the definition with an empty pattern match. Semantics of implication introduction (*Implies_I*) simply recurses on the subderivation Hp using an extended context (v', env) . Finally, we define the semantics of implication elimination as simply function application of the results of the recursive call on its two subderivations.

```

Program Fixpoint sem_nat_ded {G p}(H : nat_ded G p)
  : sem_ctx G → sem_form p :=
  match H with
  | Id v ⇒ fun env ⇒ sem_var v env
  | ExFalsum Hf ⇒ fun env ⇒
    match sem_nat_ded Hf env with
    end
  | Implies_I Hp ⇒ fun env v' ⇒ sem_nat_ded Hp (v', env)
  | Implies_E Hp Ha ⇒ fun env ⇒ (sem_nat_ded Hp env) (sem_nat_ded Ha env)
  end.

```

Using all thoses previously defined pieces we can prove the consistency of our little natural deduction system merely by showing that it should not be the case that we have a proof of *Falsum* using the empty set of assumptions. We can proof such fact by exhibiting a term of type $\text{nat_ded } \emptyset \text{ Falsum} \rightarrow \text{False}$ ¹, which is trivially done by using function *sem_nat_ded*.

Theorem *consistency* : $\text{nat_ded } \emptyset \text{ Falsum} \rightarrow \text{False} := \text{fun } p \Rightarrow \text{sem_nat_ded } p \text{ tt}.$

4. Conclusion

In this work we briefly describe a Coq formalization of a semantics based consistency proof for minimal propositional logic. The complete proof is only 85 lines long and only use of some basic dependently typed programming features of Coq. We also formalize the consistency of this simple logic in Coq using Gentzen’s admissibility of cut approach which resulted in around 270 lines of code and uses some extra proof tactics libraries. As future work, we intend to extend the current formalization to full propositional logic and also other formalisms like Hilbert systems and analytic tableaux [Smullyan 1995].

Referências

- Augustsson, L. and Carlsson, M. (1999). An exercise in dependent types: A well-typed interpreter. In *In Workshop on Dependent Types in Programming, Gothenburg*.
- Chlipala, A. (2013). *Certified Programming with Dependent Types: A Pragmatic Introduction to the Coq Proof Assistant*. The MIT Press.
- Negri, S., von Plato, J., and Ranta, A. (2001). *Structural Proof Theory*. Cambridge University Press.
- Saselli, F., Amaro, M., and Ribeiro, R. (2020). A semantical proof of consistency for minimal propositional logic in coq. <https://github.com/rodrigogribeiro/consistency-coq>.
- Smullyan, R. (1995). *First-order Logic*. Dover books on advanced mathematics. Dover.

¹Here we use the fact that $\neg\alpha$ is equivalent to $\alpha \supset \perp$.