

Syntax vs Semantics: Comparing Consistency Proofs for Minimal Propositional Logics

Felipe Sasdelli

felipe.sasdelli@aluno.ufop.edu.br

Departamento de Computação

Universidade Federal de Ouro Preto

Samuel Feitosa

samuel.feitosa@ifsc.edu.br

Departamento de Informática

Caçador, Santa Catarina, Brazil

Maycon Amaro

maycon.amaro@aluno.ufop.edu.br

Departamento de Computação

Universidade Federal de Ouro Preto

Rodrigo Ribeiro

rodrigo.ribeiro@ufop.edu.br

Programa de Pós-Graduação em Ciência da Computação

Universidade Federal de Ouro Preto

ABSTRACT

Consistency is a key property of any logical system. However, proofs of consistency usually rely on heavy proof theory notions like admissibility of cut. A more semantics-based approach to consistency proofs is to explore the correspondence between a logics and its relation with the evaluation in a λ -calculus, known as Curry-Howard isomorphism. In this work, we describe and present a comparison between two Coq formalizations of consistency of minimal propositional logic: one using a semantic-based approach and another following the traditional syntactic proof theoretical approach.

CCS CONCEPTS

• Theory of computation \rightarrow Proof theory.

KEYWORDS

Consistency, Curry-Howard Isomorphism, Coq proof assistant

ACM Reference Format:

Felipe Sasdelli, Maycon Amaro, Samuel Feitosa, and Rodrigo Ribeiro. 2020. Syntax vs Semantics: Comparing Consistency Proofs for Minimal Propositional Logics. In *Proceedings of XXIV BRAZILLIAN SYMPOSIUM ON PROGRAMMING LANGUAGES (SBLP2020)*. ACM, New York, NY, USA, 6 pages.

1 INTRODUCTION

A crucial property of a logical system is consistency, which states that it does not entails a contradiction. Basically, consistency implies that not all formulas are provable. While having a simple motivation, consistency proofs rely on the well-known admissibility of cut property, which has a rather delicate inductive proof. Gentzen, in his seminal work [6], gives the first consistency proof of logic by introducing an auxiliary formalism, the sequent calculus, in which consistency is trivial. Next, Gentzen showed that the natural

deduction system is equivalent to his sequent calculus extended with an additional rule: the cut rule. The final (and hardest) piece of Gentzen's proof is to show that the cut rule is redundant, i.e., it is admissible. As a consequence, we know something stronger: all propositions provable in the natural deduction system are also provable in the sequent calculus without cut. Since we know that the sequent calculus is consistent, we hence also know that the natural deduction calculus is [10].

However, proving the admissibility of cut is not easy, even for simple logics. Proofs of admissibility need nested inductions, and we need to be really careful to ensure a decreasing measure on each use of the inductive hypothesis. Such proofs have a heavy syntactic flavor since they recursively manipulate proof tree structures to eliminate cuts. A more semantics-based approach relies on interpreting logics as its underlying λ -calculus and proves consistency by using its computation machinery. In this work, we report on formalizing these two approaches for a minimal version of propositional logics.

This work results from a research project motivated by questions raised by undergraduate students on a first course on formal logics at Universidade Federal de Ouro Preto. The students were encouraged to "find the answer" by formalizing it in proof assistant systems. After some months following basic exercises on Agda and Coq on-line text books [11, 16], they were able to start the formalization of consistency for propositional logics. This work reports on the Coq formalization of two different approaches for consistency proofs for a minimal version of propositional logics and briefly discuss an Agda formalization of the same results also considering the conjunction and disjunction connectives. We are aware that more extensive formalizations of propositional logic already exists in Coq [15] and other proof assistants [9]. However, our focus is on showing how a better understanding of the Curry-Howard correspondence can lead to simple formalizations of mathematical results through its computational representation.

More specifically, we contribute:

- We present a semantics-based consistency proof for minimal propositional logic in Coq. Our proof is completely represented as Coq functions using dependently-typed pattern matching in less than 90 lines of code.
- We also formalize the traditional proof theoretical cut-based proof of consistency. Unlike the semantics-based proof, this formalization required the definition of several intermediate

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from permissions@acm.org.

SBLP2020, September 23–27, 2020, Natal

© 2020 Association for Computing Machinery.

ACM ISBN 978-x-xxxx-xxxx-x/YY/MM...\$15.00

definitions and lemmas to complete the proof. Instead of focusing on presenting tactic scripts, we outline the proof strategies used in the main lemmas used to ensure the consistency.

- We formalize the same results in the context of a broader version of propositional logics in Agda programming language and present some conclusions obtained by coding these results in a different proof assistant.

We organize this work as follows: Section 2 presents basic definitions about the logic considered and Section 3 presents a brief introduction to the Coq proof assistant. Section 4 describes the semantics-based proof of consistency implemented in Coq and Section 5 presents our formalization of Gentzen’s style consistency proof. We briefly discuss our Agda formalization in Section 6. Section 7 draws some lessons learned during the formalization of these consistency proofs. Finally, Section 8 presents related works and Section 9 concludes.

The complete formalization was verified using Coq version 8.10.2 and it is available on-line [12] together with the \LaTeX files needed to build this work.

2 BASIC DEFINITIONS

In this work, we consider a fragment of propositional logic which is formed by the constant *falsum* (\perp), logic implication (\supset) and variables (represented by meta-variable v), as described by the following context free grammar:

$$\alpha ::= \perp \mid v \mid \alpha \supset \alpha$$

Following common practice, we let meta-variable Γ denote contexts by a list of formulas where \emptyset denotes the empty context and $\Gamma \cup \{\alpha\}$ includes the formula α in Γ . Using contexts, we can define natural deduction as an inductively defined judgment $\Gamma \vdash \alpha$ which denotes that the formula α can be deduced from the hypothesis present in Γ using the following rules:

$$\frac{\alpha \in \Gamma}{\Gamma \vdash \alpha} \{Id\} \quad \frac{\Gamma \vdash \perp}{\Gamma \vdash \alpha} \{Ex\}$$

$$\frac{\Gamma \cup \{\alpha\} \vdash \beta}{\Gamma \vdash \alpha \supset \beta} \{\supset-I\} \quad \frac{\Gamma \vdash \alpha \supset \beta \quad \Gamma \vdash \alpha}{\Gamma \vdash \beta} \{\supset-E\}$$

Rule *Id* shows that any hypothesis in Γ is provable and rule *Ex* which specifies that from a contradiction we can deduce any formula. The rule $\supset -I$ shows that we can deduce $\alpha \supset \beta$ if we are able to prove β from $\Gamma \cup \{\alpha\}$ and rule $\supset -E$ is the well-known *modus ponens* rule.

We let notation $\Gamma \Rightarrow \alpha$ denote that α is deducible from the hypothesis in Γ using the rules of the sequent calculus which are presented next. The only difference with the natural deduction is in one rule for implication. The sequent calculus rule counter-part for implication elimination is called implication left rule, and it states that we can conclude any formula γ in a context Γ if we have that: 1) $\alpha \supset \beta \in \Gamma$; 2) $\Gamma \Rightarrow \alpha$ and 3) $\Gamma \cup \{\beta\} \Rightarrow \gamma$.

$$\frac{\alpha \in \Gamma}{\Gamma \Rightarrow \alpha} \{Id\} \quad \frac{\Gamma \Rightarrow \perp}{\Gamma \Rightarrow \alpha} \{Ex\}$$

$$\frac{\Gamma \cup \{\alpha\} \Rightarrow \beta}{\Gamma \Rightarrow \alpha \supset \beta} \{\supset-R\}$$

$$\frac{\alpha \supset \beta \in \Gamma \quad \Gamma \Rightarrow \alpha \quad \Gamma \cup \{\beta\} \vdash \gamma}{\Gamma \Rightarrow \gamma} \{\supset-L\}$$

We say that the natural deduction system is consistent if there is no proof of $\emptyset \vdash \perp$. The same idea applies to sequent calculus. In the next section, we describe our Coq formalization of the consistency for natural deduction using a semantics-based approach.

3 A TASTE OF COQ PROOF ASSISTANT

Coq is a proof assistant based on the calculus of inductive constructions (CIC) [2], a higher order typed λ -calculus extended with inductive definitions. Theorem proving in Coq follows the ideas of the so-called “BHK-correspondence”¹, where types represent logical formulas, λ -terms represent proofs [14] and the task of checking if a piece of text is a proof of a given formula corresponds to checking if the term that represents the proof has the type corresponding to the given formula.

However, writing a proof term whose type is that of a logical formula can be a hard task, even for very simple propositions. In order to make the writing of complex proofs easier, Coq provides *tactics*, which are commands that can be used to construct proof terms in a more user friendly way.

As a tiny example, consider the task of proving the following simple formula of propositional logic:

$$(A \rightarrow B) \rightarrow (B \rightarrow C) \rightarrow A \rightarrow C$$

In Coq, such theorem can be expressed as:

Section *EXAMPLE*.

Variables $A B C$: Prop.

Theorem *example* : $(A \rightarrow B) \rightarrow (B \rightarrow C) \rightarrow A \rightarrow C$.

Proof.

intros $H H' HA$. apply H' . apply H . assumption.

Qed.

End *EXAMPLE*.

In the previous source code piece, we have defined a Coq section named *EXAMPLE*² which declares variables A , B and C as being propositions (i.e. with type Prop). Tactic *intros* introduces variables H , H' and HA into the (typing) context, respectively with types $A \rightarrow B$, $B \rightarrow C$ and A and leaves goal C to be proved. Tactic *apply*, used with a term t , generates goal P when there exists $t : P \rightarrow Q$ in the typing context and the current goal is Q . Thus, apply H' changes the goal from C to B and apply H changes the goal to A . Tactic *assumption* traverses the typing context to find a hypothesis that matches with the goal.

¹ Abbreviation of Brouwer, Heyting, Kolmogorov, de Bruijn and Martin-Löf Correspondence. This is also known as the Curry-Howard “isomorphism”.

² In Coq, we can use sections to delimit the scope of local variables.

We define next a proof of the previous propositional logical formula that, in contrast to the previous proof, that was built using tactics (intros, apply and assumption), is coded directly as a function:

Definition *example'*

```
: (A → B) → (B → C) → A → C :=
fun (H : A → B) (H' : B → C) (HA : A) ⇒ H' (H HA).
```

However, even for very simple theorems, coding a definition directly as a Coq term can be a hard task. Because of this, the use of tactics has become the standard way of proving theorems in Coq. Furthermore, the Coq proof assistant provides not only a great number of tactics but also have a domain specific language for scripted proof automation, called *Ltac*. More information about Coq and *Ltac* can be found in [2, 5].

4 SEMANTICS-BASED PROOF

Our first task in formalizing the consistency is how to represent formulas (type α), which are represented by the false constant (*Falsum* constructor) and implication (*Implies* constructor). Contexts are just a list of formulas.

```
Inductive  $\alpha$  : Set :=
| Falsum :  $\alpha$ 
| Implies :  $\alpha \rightarrow \alpha \rightarrow \alpha$ .
```

Definition $\Gamma := \text{list } \alpha$.

In order to represent variables, We follow a traditional approach in programming languages community and use *De Bruijn indexes* [3] represented as an inductive judgement between formulas and contexts:

$$\frac{}{\alpha \in (\alpha :: \Gamma)} \text{ \{Here\} } \quad \frac{\alpha \in \Gamma}{\alpha \in (\beta :: \Gamma)} \text{ \{There\} }$$

In essence, this judgment states the membership of a formula α in a context Γ . The Coq encoding of this predicate is straightforward.

```
Inductive var :  $\Gamma \rightarrow \alpha \rightarrow \text{Type} :=
| Here :  $\forall G p, \text{var } (p :: G) p$ 
| There :  $\forall G p p', \text{var } G p \rightarrow \text{var } (p' :: G) p$ .$ 
```

The first constructor of type *var* specifies that a formula α is in the context $\alpha :: \Gamma$ and the constructor *There* specifies that if a formula α is in Γ , then we have $\alpha \in (\beta :: \Gamma)$, for any formula β .

Using the previous definitions, we can implement natural deduction rules for our minimal logic, as presented below.

```
Inductive nd :  $\Gamma \rightarrow \alpha \rightarrow \text{Type} :=
| Id :  $\forall G p, \text{var } G p \rightarrow \text{nd } G p$ 
| ExFalsum :  $\forall G p, \text{nd } G \text{ Falsum} \rightarrow \text{nd } G p$ 
| Implies_I :  $\forall G p p', \text{nd } (p' :: G) p \rightarrow \text{nd } G (\text{Implies } p' p)$ 
| Implies_E :  $\forall G p p', \text{nd } G (\text{Implies } p' p) \rightarrow \text{nd } G p' \rightarrow \text{nd } G p$ .$ 
```

The first rule (*Id*) establishes that any formula in the context is provable and rule *ExFalsum* defines the principle *ex-falso quodlibet*, which allows us to prove any formula if we have a deduction of *Falsum*. Rule *Implies_I* specifies that from a deduction of a formula p from a context $p' :: G$, $\text{nd } (p' :: G) p$, we can prove the implication *Implies* $p' p$. The last rule, *Implies_E*, represents the well-known *modus-ponens*, which allows us to deduce a formula p from deductions of *Implies* $p' p$ and p' .

The Curry-Howard isomorphism states that there is a correspondence between logics and functional programming by relating logical formulas to types and proofs to λ -calculus terms [14]. In order to prove consistency of natural deduction system, we use this analogy with λ -calculus. Basically, under the Curry-Howard interpretation, saying that there is no proof for $\emptyset \vdash \perp$ (the statement of the consistency property) resorts to show that there is no value³ of type \perp . A way to ensure that a type has no value, is to reduce arbitrary terms until we no more reductions steps apply and that is the strategy of our semantics-based proof: build an algorithm to reduce proof terms and use it to show that there are no proofs for \perp .

The reduction algorithm we use is an well-typed interpreter for the simply-typed λ -calculus based on a standard model construction. The first step in the implementation is to define the denotation of a formula by recursion on its structure. The idea is to associate the empty type (*False*) with the formula *Falsum* and a function type with formula *Implies* $p1 p2$, as presented next.

```
Program Fixpoint sem_form (p :  $\alpha$ ) : Type :=
match p with
| Falsum ⇒ False
| Implies p1 p2 ⇒ sem_form p1 → sem_form p2
end.
```

Using *sem_form* function, we can define context semantics as tuples of formula semantics as follows:

```
Program Fixpoint sem_ctx (G :  $\Gamma$ ) : Type :=
match G with
|  $\emptyset$  ⇒ unit
| (t :: G') ⇒ sem_form t × sem_ctx G'
end.
```

Function *sem_ctx* recurses over the structure of the input context building right-nested tuple ending with the Coq *unit* type, which is a type with a unique element. Since contexts are mapped into tuples, variables must be mapped into projections on such tuples. This would allow us to retrieve the value associated with a variable in a context.

```
Program Fixpoint sem_var {G p}(v : var G p)
: sem_ctx G → sem_form p :=
match v with
| Here ⇒ fun env ⇒ fst env
| There v' ⇒ fun env ⇒ sem_var v' (snd env)
end.
```

Function *sem_var* receives a variable (value of type *var* $G p$) and a semantics of a context (a value of type *sem_ctx* G) and returns the value of the formula represented by such variable. Whenever the variable is built using constructor *Here*, we just return the first component of the input context semantics, and when we have the constructor *There*, we just call *sem_var* recursively.

Our next step is to define the semantics of natural deduction proofs. The semantics of proofs is done by function *sem_nat_ded*, which maps proofs (values of type *nat_ded* $G p$) and context semantics (values of type *sem_ctx* G) to the value of input proof conclusion (type *sem_form* p). The first case specifies that the semantics of an

³A value is a well-typed term which can not be further reduced according to a semantics.

identity rule proof (constructor *Id*) is just retrieving the value of the underlying variable in the context semantics by calling function *sem_var*. The second case deals with *ExFalsum* rule: we recurse over the proof object *Hf* which will produce a Coq object of type *False*, which is empty and so we can finish the definition with an empty pattern match. Semantics of implication introduction (*Implies_I*) simply recurses on the subderivation *Hp* using an extended context (v' , *env*). Finally, we define the semantics of implication elimination as simply function application of the results of the recursive call on its two subderivations.

```

Program Fixpoint sem_nat_ded {G p} (H : nat_ded G p)
  : sem_ctx G → sem_form p :=
  match H with
  | Id v ⇒ fun env ⇒ sem_var v env
  | ExFalsum Hf ⇒ fun env ⇒
    match sem_nat_ded Hf env with end
  | Implies_I Hp ⇒ fun env v' ⇒ sem_nat_ded Hp (v', env)
  | Implies_E Hp Ha ⇒ fun env ⇒
    (sem_nat_ded Hp env) (sem_nat_ded Ha env)
  end.

```

Using all those previously defined pieces, we can prove the consistency of our little natural deduction system merely by showing that it should not be the case that we have a proof of *Falsum* using the empty set of assumptions. We can prove such fact by exhibiting a term of type $\text{nat_ded } \emptyset \text{ Falsum} \rightarrow \text{False}$ ⁴, which is trivially done by using function *sem_nat_ded* with term *tt*, which is the value of type *unit* that denotes the semantics of the empty context needed to call *sem_nat_ded*.

```

Theorem consistency : nat_ded  $\emptyset$  Falsum → False
  := fun p ⇒ sem_nat_ded p tt.

```

5 GENTZEN STYLE PROOF

Now, we turn our attention to formalizing the admissibility of cut based consistency proof in Coq. Unlike our semantics-based proof, which uses dependently typed syntax to concisely represent formulas and natural deduction proofs, we use an explicitly approach in representing formulas as sequent calculus proofs. We use natural numbers to represent variables and formulas are encoded as simple inductive type which has an immediate meaning.

Definition *var* := \mathbb{N} .

```

Inductive  $\alpha$  : Type :=
| Falsum :  $\alpha$ 
| Var : var →  $\alpha$ 
| Implies :  $\alpha \rightarrow \alpha \rightarrow \alpha$ .

```

The main change on how we represent the sequent calculus is in the rule for variables. We use the Coq library boolean list membership predicate *member*, which fits better for proof automation. In order to simplify the task of writing code that uses this predicate, we defined notation $a \in G$ which means *member a G*.

Next, we the sequent calculus formulation for our minimal logic. The only difference with the natural deduction is in one rule for implication. The sequent calculus rule counter-part for implication elimination is called implication left rule, which states that we can

⁴Here we use the fact that $\neg\alpha$ is equivalent to $\alpha \supset \perp$.

conclude any formula γ in a context Γ if we have that: 1) $\alpha \supset \beta \in \Gamma$; 2) $\Gamma \Rightarrow \alpha$ and 3) $\Gamma \cup \{\beta\} \Rightarrow \gamma$. The rules for the sequent-calculus and its correspondent Coq implementation are presented next.

```

Inductive seq_calc :  $\Gamma \rightarrow \alpha \rightarrow \text{Prop}$  :=
| Id G a
  : (Var a) ∈ G → seq_calc G (Var a)
| Falsum_L G a
  : Falsum ∈ G → seq_calc G a
| Implies_R G a b
  : seq_calc (a :: G) b →
    seq_calc G (Implies a b)
| Implies_L G a b c
  : (Implies a b) ∈ G →
    seq_calc G a →
    seq_calc (b :: G) c →
    seq_calc G c.

```

An important property of sequent calculus derivations is the weakening which states that it stable under the inclusion of new hypothesis.

Lemma 1 (Weakening). If $\Gamma \subseteq \Gamma'$ and $\Gamma \Rightarrow \alpha$ then $\Gamma' \Rightarrow \alpha$.

PROOF. Induction on the derivation of $\Gamma \Rightarrow \alpha$. \square

Since weakening has a straightforward inductive proof (coded as 4 lines tactic script), we do not comment on its details. However, this proof is used in several points in the admissibility of cut property, which we generalize using the following lemma in order to get a stronger induction hypothesis.

Lemma 2 (Generalized admissibility). If $\Gamma \Rightarrow \alpha$ and $\Gamma' \Rightarrow \beta$ then $\Gamma \cup (\Gamma' - \{\alpha\}) \Rightarrow \beta$.

PROOF. The proof proceeds by induction on the structure of the cut formula α . The cases for $\alpha = \perp$ and α is a variable easily follows by induction on $\Gamma \Rightarrow \alpha$ and using weakening on the variable case. The interesting case is when $\alpha = \alpha_1 \supset \alpha_2$ in which we proceed by induction on $\Gamma \Rightarrow \alpha$. Again, most of cases are straightforward except when the last rule used to conclude $\alpha_1 \supset \alpha_2$ was $\supset -R$. In this situation, we proceed by induction on $\Gamma' \Rightarrow \beta$. The only interesting cases are when the last rule was $\supset -L$ or $\supset -R$. If the last rule used in deriving $\Gamma' \Rightarrow \beta$ was $\supset -R$ we have: $\beta = a \supset b$, for some a, b . Also, we have that $\Gamma' \cup \{a\} \Rightarrow b$. By the induction hypothesis on $\Gamma' \cup \{a\} \Rightarrow b$, we have that $\Gamma \cup ((\Gamma' \cup \{a\}) - \{\alpha_1 \supset \alpha_2\}) \Rightarrow b$. Since we have $\Gamma \cup ((\Gamma' \cup \{a\}) - \{\alpha_1 \supset \alpha_2\}) \Rightarrow b$ then we also have $\Gamma \cup (\Gamma' \cup \{a\} - \{\alpha_1 \supset \alpha_2\}) \cup \{a\} \Rightarrow b$ and conclusion follows by rule $\supset -R$. The case for $\supset -L$ follows the same structure. \square

Using the previous defined lemma, the admissibility of cut is an immediate corollary.

Corollary 1 (Admissibility of cut). If $\Gamma \Rightarrow \alpha$ and $\Gamma \cup \{\alpha\} \Rightarrow \beta$ then $\Gamma \Rightarrow \beta$.

PROOF. Immediate consequence of Lemmas 1 and 2. \square

Consistency of sequent calculus trivially follows by inspection on the structure of derivations.

Theorem 1 (Consistency of sequent calculus). There is no proof of $\emptyset \Rightarrow \perp$.

PROOF. Immediate from the sequent calculus rules (there is no rule to introduce \perp). \square

The next step in the mechanization of the consistency of our minimal logic is to stabilish the equivalence between sequent calculus and natural deduction systems. The equivalence proofs between these two formalism are based on a routine induction on derivations using admissibility of cut. We omit its description for brevity. The complete proofs of these equivalence results can be found in our code repository [12]. Finally, we can prove the consistency of natural deduction by combining the proofs of consistency of the sequent calculus and the equivalence between these formalisms.

Theorem 2 (Consistency for Natural Deduction). There is no proof of $\emptyset \vdash \perp$.

PROOF. Suppose that $\emptyset \vdash \perp$. By the equivalence between natural deduction and sequent calculus, we have $\emptyset \Rightarrow \perp$, which contradicts Theorem 1. \square

6 AGDA FORMALIZATION

In this section we briefly present some details of our Agda formalization of consistency proofs for propositional logics. Since the Agda version of the consistency proof using a well-typed interpreter for the simply-typed λ -calculus is essentially the same as our Coq implementation, we will focus on the admissibility of cut version.

One important design decision of our Agda proof was how to represent a permutation relation between contexts. In our Coq code we simply “lift” the boolean list membership test to set equality relation and use facilities offered by small-scale reflection and type classes to ease evidence construction [7, 8].

7 LESSONS LEARNED

Previous sections presented two different formalizations for consistency of minimal propositional logic in Coq proof assistant. In this section, we briefly resume the main characteristics of each approach and try to draw some conclusions on the realized proof effort.

The first strategy is inspired by the Curry-Howard correspondence and it is, in essence, a well-typed interpreter for the simply-typed λ -calculus. The consistency is ensured by construct a term that asserts that is impossible to build a term of type *Falsum* from an empty context, which is done by a simple call to the λ -calculus interpreter. The complete formalization is 85 lines long and we only use the *Program* construct which eases the task of dependently-typed pattern matching, which is necessary to construct functions which manipulate richly typed structures like type *var* or build types from values like *sem_form* and *sem_ctx*. No standard tactic or tatic library is used to finish the formalization.

The second strategy implements the usual proof theoretical approach to guarantee the consistency of logics. As briefly described in the previous section, proving the admissibility of cut needs nested inductions on the structure of the cut-formula and on the structure of the sequent-calculus derivations. The main problem on proving the cut lemma is the bureocratic adjustment of contexts by using weakening in the right places in the proof. Our proof uses some

tactics libraries [5, 11] and type class based automation to automatically produce proof terms for subset relation between contexts. Our cut-based consistency proof has around 270 lines of code without considering the tactics libraries used.

When comparing the both approaches, it is obvious that the second demands approximately 3 times more lines of code than the first. However, while demanding more code, the cut-based proof essentially follows the ideas used in proof-theory textbooks. One of main difficulties in formalizing the Gentzen style proof was the correct handling of weakening. The usage of proof automation tools and Coq type classes had a great impact on the simplification of these results. The semantics-based proof rely on the relation between the minimal propositional logic and the simply-typed λ -calculus, i.e., it is necessary to understand the consequences of the Curry-Howard isomorphism.

8 RELATED WORK

Formalizations of logics. Proof assistants has been used with sucess to formalize several logical theories. van Doorn describes a formalization of some important results about propositional logic in Coq: completeness of natural deduction, equivalence between natural deduction and sequent calculus and the admissibility of cut theorem [15]. In his formalization, van Doorn considered the full syntax of propositional logic (including negation, disjunction and conjunction) and also have proved completeness of natural deduction. In our work, we tried to keep things to a bare minimum by considering a minimalistic version of propositional logic. We intend to include the missing conectives as future work. Another formalization of propositional logic was implemented by Michaelis and Nipkow [9] which covered several proof systems (sequent calculus, natural deduction, Hilbert systems, resolution) and proved the some important meta-theoretic results like: compactness, translations between proof systems, cut-elimination and model existence.

A formalization of linear logic was conducted by Allais and McBride [1]. In essence, Allais and McBride work starts from a well-scoped λ -calculus and introduce a typed representation which leads to a intuitionistic version of linear logic which uses a relation that ensure the resource control behavior of linear logic proofs. Another work which formalizes linear logic was developed by Xavier et. al [17]. The main novelty of their work was the formalization of a focused linear logic using a binding representation called parametric high-order abstract syntax (PHOAS) [4].

Applications of proof assistants. Ribeiro and Du Bois [?] described the formalization of a RE (regular expression) parsing algorithm that produces a bit representation of its parse tree in the dependently typed language Agda. The algorithm computes bit-codes using Brzozowski derivatives and they proved that the produced codes are equivalent to parse trees ensuring soundness and completeness with respect to an inductive RE semantics. They included the certified algorithm in a tool developed by themselves, named verigrep, for RE-based search in the style of GNU grep. While the authors provided formal proofs, their tool show a bad performance when compared to other approaches to RE parsing.

A formal constructive theory of RLs (regular language) was presented by Doczkal et. al. in [?]. They formalized some fundamental results about RLs. For their formalization, they used the Ssreflect

extension to Coq, which features an extensive library with support for reasoning about finite structures such as finite types and finite graphs. They established all of their results in about 1400 lines of Coq, half of which are specifications. Most of their formalization deals with translations between different representations of RLs, including REs, DFAs (deterministic finite automata), minimal DFAs and NFAs (non-deterministic finite automata). They formalized all these (and other) representations and constructed computable conversions between them. Besides other interesting aspects of their work, they proved the decidability of language equivalence for all representations. Unlike our work, Doczkal et. al.'s only concerns about formalizing classical results of RL theory in Coq, without using the formalized automata in practical applications, like matching or parsing.

9 CONCLUSION

In this work we briefly describe a Coq formalization of a semantics based consistency proof for minimal propositional logic. The complete proof is only 85 lines long and only use of some basic dependently typed programming features of Coq. We also formalize the consistency of this simple logic in Coq using Gentzen's admissibility of cut approach which resulted in longer formalization: the formalization has around 270 lines of code, which were much simplified by using some tactics libraries.

As future work, we intend to extend the current formalization to full propositional logic and also other formalisms like Hilbert systems and analytic tableaux [13].

REFERENCES

- [1] Guillaume Allais. 2018. Typing with Leftovers - A mechanization of Intuitionistic Multiplicative-Additive Linear Logic. In *23rd International Conference on Types for Proofs and Programs (TYPES 2017) (Leibniz International Proceedings in Informatics (LIPIcs))*, Andreas Abel, Fredrik Nordvall Forsberg, and Ambrus Kaposi (Eds.), Vol. 104. Schloss Dagstuhl-Leibniz-Zentrum fuer Informatik, Dagstuhl, Germany, 1:1–1:22. <https://doi.org/10.4230/LIPIcs.TYPES.2017.1>
- [2] Yves Bertot and Pierre Castéran. 2004. *Interactive Theorem Proving and Program Development. Coq'Art: The Calculus of Inductive Constructions*. Springer Verlag.
- [3] N.G. Bruijn, de. 1972. Lambda calculus notation with nameless dummies, a tool for automatic formula manipulation, with application to the Church-Rosser theorem. *Indagationes Mathematicae (Proceedings)* 75, 5 (1972), 381–392. [https://doi.org/10.1016/1385-7258\(72\)90034-0](https://doi.org/10.1016/1385-7258(72)90034-0)
- [4] Adam Chlipala. 2008. Parametric Higher-Order Abstract Syntax for Mechanized Semantics. In *Proceedings of the 13th ACM SIGPLAN International Conference on Functional Programming (ICFP 08)*. Association for Computing Machinery, New York, NY, USA, 143–156. <https://doi.org/10.1145/1411204.1411226>
- [5] Adam Chlipala. 2013. *Certified Programming with Dependent Types: A Pragmatic Introduction to the Coq Proof Assistant*. The MIT Press.
- [6] Gerhard Gentzen. 1936. Die Widerspruchsfreiheit der reinen Zahlentheorie. *Math. Ann.* 112 (1936), 493–565.
- [7] Georges Gonthier and Assia Mahboubi. 2010. An introduction to small scale reflection in Coq. *J. Formalized Reasoning* 3, 2 (2010), 95–152. <https://doi.org/10.6092/issn.1972-5787/1979>
- [8] Georges Gonthier, Beta Ziliani, Aleksandar Nanevski, and Derek Dreyer. 2011. How to make ad hoc proof automation less ad hoc. In *Proceeding of the 16th ACM SIGPLAN international conference on Functional Programming, ICFP 2011, Tokyo, Japan, September 19-21, 2011*, Manuel M. T. Chakravarty, Zhenjiang Hu, and Olivier Danvy (Eds.). ACM, 163–175. <https://doi.org/10.1145/2034773.2034798>
- [9] Julius Michaelis and Tobias Nipkow. 2017. Propositional Proof Systems. *Archive of Formal Proofs* (June 2017). http://isa-afp.org/entries/Propositional_Proof_Systems.html, Formal proof development.
- [10] Sara Negri, Jan von Plato, and Aarne Ranta. 2001. *Structural Proof Theory*. Cambridge University Press. <https://doi.org/10.1017/CBO9780511527340>
- [11] Benjamin C. Pierce, Arthur Azevedo de Amorim, Chris Casinghino, Marco Gaboardi, Michael Greenberg, Catalin Hritcu, Vilhelm Sjöberg, and Brent Yorgey. 2018. *Logical Foundations*. Electronic textbook. Version 5.5. <http://www.cis.upenn.edu/~bcpierce/sf>.
- [12] Felipe Sasdeli, Maycon Amaro, and Rodrigo Ribeiro. 2020. A Semantical Proof of Consistency for Minimal Propositional Logic in Coq. Available at <https://github.com/rodrigoribeiro/consistency-coq>.
- [13] R.M. Smullyan. 1995. *First-order Logic*. Dover. <https://books.google.com.br/books?id=kgvvhQ-oSZiUC>
- [14] Morten Heine Sørensen and Pawel Urzyczyn. 2006. *Lectures on the Curry-Howard Isomorphism, Volume 149 (Studies in Logic and the Foundations of Mathematics)*. Elsevier Science Inc., USA.
- [15] Floris van Doorn. 2015. Propositional Calculus in Coq. arXiv:math.LO/1503.08744
- [16] Philip Wadler and Wen Kokke. 2019. *Programming Language Foundations in Agda*. Available at <http://plfa.inf.ed.ac.uk/>.
- [17] Bruno Xavier, Carlos Olarte, Giselle Reis, and Vivek Nigam. 2017. Mechanizing Focused Linear Logic in Coq. In *12th Workshop on Logical and Semantic Frameworks, with Applications, LSEA 2017, Brasília, Brazil, September 23-24, 2017 (Electronic Notes in Theoretical Computer Science)*, Sandra Alves and Renata Wasserman (Eds.), Vol. 338. Elsevier, 219–236. <https://doi.org/10.1016/j.entcs.2018.10.014>